

Institut des Carrières Commerciales

**Rue de la Fontaine, 4
1000 Bruxelles**

***Langage* PASCAL**

**Support du Cours
« Bases de Programmation »**

Syllabus destiné aux
1ères années de
Graduat en Informatique

Auteur :

Patrick TRAU
Université de Strasbourg

1998 - 1999

Langage PASCAL

[Patrick TRAU](#) Janvier 92

version HTML mars 97 : <http://www-ipst.u-strasbg.fr/pat/program/pascal.htm>

Ce document présente (de manière claire je crois, d'après ce qu'en ont dit certains) le langage PASCAL. Il a été publié dans APTEP - INFO il y a quelques années. Depuis, bien que je maîtrisais bien ce langage et avais développé de nombreux programmes en Pascal, je suis passé au C. Le C est d'après moi plus souple et plus efficace à condition de bien en comprendre les mécanismes (il faut d'après moi comprendre à la fois l'assembleur et les objets pour bien programmer en C). De plus le C est plus utilisé dans mon environnement d'informaticiens. Néanmoins le Pascal reste bien meilleur pour le débutant et le programmeur moyen : parfaitement structuré et clair, il conduit rapidement à un programme de bonne qualité et assez facilement maintenable (ce qui l'est moins en C, qui possède trop souvent trop de possibilités pour résoudre le même problème).

De plus, on trouve des compilateurs Pascal très conviviaux (Turbo Pascal sur PC, avec aide en ligne, déboguage,...).

Ce document reste donc d'actualité : le Pascal est d'après moi une très bonne manière d'aborder la programmation. C'est pourquoi j'ai choisi de mettre à disposition ce document sur Internet, qu'il serve à qui en a besoin ! Néanmoins, je n'ai pas passé trop de temps à sa mise en page, les graphiques restent en mode caractère, il n'y a pas de lien direct entre les sujets d'exercices et leur correction (que l'on trouvera en fin du document).

Voici quelques adresses intéressantes :

- [Pascalissime](http://www.mygale.org/~jmy51) (<http://www.mygale.org/~jmy51>)
- [Point de rencontre des utilisateurs de Borland Pascal](http://members.aol.com/chaussaron/bcpasc.htm) (<http://members.aol.com/chaussaron/bcpasc.htm>)
- [Site officiel de Borland France](http://www.borland.fr) (<http://www.borland.fr>)

Copyright : utilisation de ce document libre pour tout usage personnel. Utilisation autorisée pour tout usage public non commercial, à condition de citer son auteur ([Patrick TRAU](#), [IPST](#), [Université Louis Pasteur Strasbourg](#)) et de me signaler tout usage intensif. Utilisation commerciale interdite sans accord écrit de ma part.

Table des Matières

PASCAL	1
LANGAGE PASCAL	2
TABLE DES MATIERES	3
INTRODUCTION	4
<i>LES LOGICIELS</i>	4
<i>ORGANISATION DE L'ORDINATEUR</i>	4
<i>LANGAGES DE PROGRAMMATION</i>	4
UN PREMIER PETIT PROGRAMME	6
CONSTANTES	8
INSTRUCTION D'AFFECTATION	9
LES TYPES DE VARIABLES STANDARD SIMPLES ET OPERATEURS ASSOCIES	10
<i>ENTIERS</i>	10
<i>REELS</i>	10
<i>BOOLEENS</i>	10
<i>CARACTERES</i>	11
LES FONCTIONS STANDARD	12
INSTRUCTION	13
STRUCTURES DE CONTROLE	14
<i>BOUCLE WHILE - DO (tant que - faire)</i>	14
<i>BOUCLE REPEAT - UNTIL (répéter - jusqu'à ce que)</i>	14
<i>BOUCLE FOR - DO (pour - faire)</i>	15
<i>INSTRUCTION IF - THEN - ELSE (si - alors - sinon)</i>	15
<i>LA STRUCTURE CASE - OF (cas - parmi)</i>	16
TYPES ENUMERES NON STANDARDS	17
LES TYPES INTERVALLES	18
TABLEAUX	19
<i>TABLEAUX UNIDIMENSIONNELS</i>	19
<i>LES CHAINES DE CARACTERES</i>	19
<i>TABLEAUX DE TABLEAUX</i>	20
<i>TABLEAUX COMPACTES</i>	20
ENREGISTREMENTS	21
<i>DECLARATION</i>	21
<i>UTILISATION DES ENREGISTREMENTS</i>	21
<i>LA STRUCTURE WITH - DO (avec - faire)</i>	22
<i>ENREGISTREMENTS AVEC VARIANTES</i>	22
PROCEDURES ET FONCTIONS	24
<i>GENERALITES</i>	24
<i>PORTEE DES DECLARATIONS</i>	25
<i>ARGUMENTS (OU PARAMETRES)</i>	26
<i>LES FONCTIONS</i>	26
<i>RECURSIVITE</i>	27
LES ENTREES / SORTIES	29
<i>SUR LA CONSOLE</i>	29
<i>LES FICHIERS DE TEXTE</i>	30
<i>EXTENSIONS NON STANDARD</i>	31
ENSEMBLES	32
POINTEURS	33
<i>LES LISTES CHAINEES ET ARBRES</i>	33
<i>LES POINTEURS EN PASCAL</i>	34
CORRECTION DES EXERCICES	36
QUI SUIS-JE ?	48

INTRODUCTION

- [LES LOGICIELS](#)
- [ORGANISATION DE L'ORDINATEUR](#)
- [LANGAGES DE PROGRAMMATION](#)

LES LOGICIELS

Dans la majorité des cas, on achète des programmes (logiciels) tout faits qui correspondent plus ou moins au besoin :

- * Traitement de texte - P.A.O :avec mise en page, justification, numérotation chapitres - pages, table des matières, dictionnaire...
- * Tableur : tableau de nombres à 2 dimensions et calculs
- * Base de données : ensemble de fiches (nom, adresse...) et recherche par rubrique, publipostage...
- * C.A.O, Dessin par ordinateur : propre, modification aisée, archivage...
- * Gestion : paye, facturation, stock...
- * Communication : transfert de programmes par modem et ligne téléphonique, serveur minitel...

Un Intégré regroupe plusieurs de ces possibilités.

Soit on achète un logiciel général : très bon niveau, parfaitement testé, documentation, formation... mais trop général (fonctions inutiles, fonctions utiles avec trop de paramètres ou difficilement accessibles). Soit on fait (ou fait faire) un logiciel particulier : plus pratique, mais plus hasardeux (erreurs, SAV, doc...). Le cahier des charges doit être très précis.

ORGANISATION DE L'ORDINATEUR

- * Multiposte : plusieurs consoles sur un même ordinateur (CPU puissant, tout est partageable)
- * Réseau : plusieurs CPU et MC non partageable (sauf réseau de multipostes), MdM et périphériques partageables ou locaux.

LANGAGES DE PROGRAMMATION

Un ordinateur est une machine bête, ne sachant qu'obéir, et à très peu de choses :

- * addition, soustraction, multiplication en binaire, uniquement sur des entiers,
- * sortir un résultat ou lire une valeur binaire (dans une mémoire par exemple),
- * comparer des nombres.

Sa puissance vient du fait qu'il peut être PROGRAMME, c'est à dire que l'on peut lui donner, à l'avance, la séquence (la suite ordonnée) des ordres à effectuer l'un après l'autre. Le grand avantage de l'ordinateur est sa rapidité. Par contre, c'est le programmeur qui doit TOUT faire. L'ordinateur ne comprenant que des ordres codés en binaire (le langage machine), des langages dits "évolués" ont été mis au point pour faciliter la programmation.

Langage PASCAL

Le PASCAL, créé par WIRTH au début des années 70, possède des instructions assez claires (si vous comprenez l'anglais), et favorise une approche méthodique et disciplinée (on dit "structurée").

Le PASCAL est un langage compilé, c'est à dire qu'il faut :

- * entrer un texte dans l'ordinateur (à l'aide d'un programme appelé EDITEUR),
- * le traduire en langage machine (c'est à dire en codes binaires compréhensibles par l'ordinateur) : c'est la compilation et éventuellement l'édition de liens (LINK),
- * l'exécuter.

Contrairement à un basic interprété, l'exécution sera beaucoup plus rapide puisqu'il n'y a plus de traduction à effectuer.

Bien que le langage soit normalisé, un certain nombre de points dépendent de la machine et du compilateur utilisé (par exemple comment appeler le compilateur). Ces indications ne seront pas données ici. Si vous avez le choix, je vous conseille TURBO PASCAL, le plus pratique d'emploi (en particulier parce qu'il possède son propre éditeur de texte).

UN PREMIER PETIT PROGRAMME

Un programme PASCAL est composé d'une entête, des déclarations et des instructions (délimitées par BEGIN et END.).

```
ex :   PROGRAM cercle (input,output);      (* entête *)
        VAR perimetre,diametre : REAL;      (* déclarations *)
        BEGIN
            readln(diametre);      (* instruction *)
            perimetre := 3.141592 * diametre; (* instruction *)
            writeln(diametre,perimetre)      (* instruction *)
        END.
```

L'entête est composée du mot PROGRAM, suivi du nom du programme (cercle), et d'indications sur les Entrées/Sorties (ici le clavier et l'écran).

La partie déclarative de notre programme est limitée à la déclaration de deux variables (mot clef VAR). Une variable est une "case" mémoire de l'ordinateur, à laquelle on donne ici un nom. Chaque case peut contenir une valeur. On a précisé ici que nos deux variables PERIMETRE et DIAMETRE contiendraient des réels. Les types simples connus en PASCAL sont : REAL, INTEGER (entier naturel), CHAR (contient UN est un seul caractère), et BOOLEAN (booléen, c.a.d qui peut valoir soit TRUE (vrai) soit FALSE (faux). En TURBO PASCAL, les entiers admissibles sont compris entre -32768 et +32767. Dans tous les PASCALS on possède la variable prédéclarée MAXINT qui donne le plus grand entier admissible. Les réels doivent être compris en TURBO entre + et -1.7E37 (c.a.d 1,7 fois 10 puissance 37), avec 11 chiffres significatifs. La virgule décimale est toujours représentée par un point en informatique.

Un identificateur (tout nom que vous choisirez : variable, programme...) peut être formé de lettres (A à Z), de chiffres et (pas sur toutes les versions de PASCAL) du signe _ (souligné). TURBO PASCAL accepte des noms de 127 caractères maximum, certains PASCAL sont plus limités (31 caractères par ex). Le premier caractère doit être une lettre. Par exemple, VALEUR1 ou PREM_VALEUR sont possibles mais pas 1ERE_VALEUR. En PASCAL les minuscules sont traitées comme des majuscules (SURface et surFACE désignent la même case mémoire). Je n'utilise les majuscules que pour faire ressortir les mots importants. Les accents et autres ç ne sont pas autorisés (var diamètre:real est interdit à cause de l'accent). Un blanc dans un identificateur est également interdit (utilisez _ pour séparer des mots dans un même identificateur).

Toute variable utilisée dans un programme doit être déclarée. Ceci évite la plupart des erreurs de frappe, et rend le programme plus compréhensible.

Les instructions de notre programme sont :

* lecture sur le clavier :

le programme s'arrête, attend que l'on donne une valeur à l'aide du clavier, met cette valeur dans la case DIAMETRE et continue le programme lorsque l'on appuie sur la touche "ENTREE" ou "RETURN".

* calcul et affectation :

on multiplie le contenu de la case DIAMETRE par PI, et on met le résultat dans la case PERIMETRE.

Le := symbolise une flèche à gauche. Ce n'est PAS une égalité au sens mathématique, mais la copie d'une valeur dans une mémoire.

* écriture sur l'écran :

Langage PASCAL

on affiche sur l'écran le contenu des case DIAMETRE (que l'on connaissait puisque nous l'avions donné) et PERIMETRE (qui nous intéresse un peu plus). Ce programme affiche donc deux chiffres. Il serait plus parlant d'afficher également des commentaires par l'instruction :

```
writeln('Diamètre : ',diameter,' Périimètre : ',perimetre)
```

Les textes doivent être entourés de cotes ('). Les majuscules/minuscules sont significatives. Pour afficher une apostrophe utiliser deux cotes ('l'exemple'). Pour sauter une ligne utiliser WRITELN seul.

Les instructions doivent toujours être séparées par des ";" (j'ai dit "séparées", pas "terminées"). Le fait de passer à la ligne n'est interprété par l'ordinateur que comme un blanc. On aurait donc pu écrire notre programme sur une seule ligne (peut-être un peu longue pour l'éditeur). Le programme doit toujours se terminer par un point.

On peut insérer des remarques dans le programme (qui ne seront pas lues par le compilateur) en les entourant par (* et *) ou { et }. On ne peut en standard pas imbriquer des commentaires. Les commentaires peuvent faire plus d'une ligne, ceci permet de supprimer momentanément une partie d'un programme.

Chaque identificateur est séparé du suivant par un blanc. On considère comme un blanc : plusieurs blanc, un retour à la ligne, un commentaire.

EXERCICE (ex_tva) Faire un programme demandant le prix unitaire HT d'un article et sa quantité, puis qui affiche : le total Hors Taxes, le montant de la TVA (pour un taux de 18,6 %) et le total TTC.

CONSTANTES

Nous avons déjà utilisé des VARIABLES (on donne un nom à une mémoire). Mais on peut également utiliser des CONSTANTES, qui sont des valeurs qui restent fixes tout au long du programme et pour chaque exécution. On déclare les constantes avant de déclarer les variables, par :
CONST nom=valeur

ex : CONST taux_tva=18.6

MAXINT (plus grand entier possible) est une constante prédéfinie.

On peut également avoir des constantes de type "chaîne de caractères" :

```
PROGRAM bonjour(output);    { on n'a pas besoin ici du clavier }
CONST message1='salut, ça va ?';
      message2='moi ça baigne';

BEGIN
  writeln(message1);
  writeln(message2)        (* notez l'absence de ; devant le END *)
END.
```

Le type de la constante est déterminé automatiquement par le compilateur (entier si nombre sans point, réel si nombre avec point, caractères si entre cotes).

Rq1: on peut toujours utiliser un entier à la place d'un réel mais pas l'inverse.

Rq2: si on utilise un point décimal, il FAUT l'encadrer de chiffres (0.3 et non .3 , 10 ou 10.0 et non 10.)

INSTRUCTION D'AFFECTATION

On appelle AFFECTATION la mise d'une valeur dans une variable. Celle-ci peut être sous forme directe ($A:=B$) ou sous forme d'un calcul ($A:=B*C$). Le signe $:=$ représente une flèche à gauche, et signifie "mettre la VALEUR à droite du $:=$ dans la mémoire désignée à gauche" (mettre le contenu de B dans A ou mettre le résultat du calcul (contenu de B) fois (contenu de C) dans A). Une affectation du type $B*C:=A$ est donc IMPOSSIBLE.

Une affectation ne peut se faire qu'entre une variable et une expression de même type (si A est réel, impossible de faire $A:='xyz'$). La seule exception est de mettre un entier dans un réel (le .0 est rajouté automatiquement), mais l'inverse est impossible directement. Soient I entier et X réel, pour mettre X dans I il faut utiliser $I:=\text{ROUND}(X)$ (arrondi) ou $I:=\text{TRUNC}(X)$ (partie entière).

On peut également utiliser des affectations et expressions booléennes. Par exemple, en ayant déclaré :

```
VAR test:boolean;  
a,b:real;
```

on peut écrire : $\text{TEST} := (A < B) \text{ and } (A > 0)$

LES TYPES DE VARIABLES STANDARD SIMPLES ET OPERATEURS ASSOCIES

- [ENTIERS](#)
- [REELS](#)
- [BOOLEENS](#)
- [CARACTERES](#)

Tous ces types, excepté les réels, sont dits "énumérés".

ENTIERS

déclaration : VAR variable1,variable2,...,variableN : INTEGER;

opérations sur entiers : + - * div (division) mod (reste de la division). Elles sont toutes à résultat entier, et nécessitent deux arguments entiers.

Les entiers sont compris entre -MAXINT et +MAXINT qui est une constante standard prédéfinie (sa valeur dépend par contre du compilateur, 32767 en TURBO).

REELS

déclaration : VAR liste de variables : REAL;

opérations : + - * /

Quand une opération comprend un argument réel et un entier, le résultat est réel. / donne toujours un résultat réel, même si les deux arguments sont entiers.

* et / sont de priorité supérieure à + et -, mais entre * et / tout dépend du compilateur (en général de gauche à droite). En cas d'ambiguïté, utilisez des parenthèses (il n'y a aucun inconvénient à mettre plus de parenthèses que nécessaire).

Exemples d'expressions numériques (soit A=3, B=4, C=2):

$A+B/C = A+(B/C) = 5$

$(A+B)/C = 3.5$

$A/B*C = (A/B)*C$ (1.5) dans certains cas, $A/(B*C)$ (0.375) dans d'autres

A/BC = valeur de A sur valeur de la variable de nom BC et non A sur $B*C$

$B*A-5*C = (B*A)-(5*C) = 2$

BOOLEENS

déclaration : VAR liste de variables : BOOLEAN;

Ces variables peuvent prendre soit la valeur TRUE (vrai), soit la valeur FALSE (faux).

opérations booléennes : AND, OR, NOT, et (non standard) XOR (ou exclusif). Ces opérations nécessitent des arguments booléens.

Langage PASCAL

opérations à valeur booléenne : > (supérieur), < (inf), >= (sup ou égal), <=, = (égal), <> (différent). Ces opérations comparent tous éléments de type simple (les 2 arguments doivent être de même type, sauf entiers et réels qui peuvent être comparés entre eux), et renvoient un booléen. Les caractères sont comparés suivant l'ordre du code ASCII.

AND (et), OR (ou), NOT (non), sont de priorité supérieure aux précédents et ne peuvent opérer que sur des booléens : A>B et C doit être écrit : (A>B) and (A>C). Les parenthèses sont obligatoires pour ne pas faire en premier B and A.

CARACTERES

déclaration : VAR liste de variables : CHAR;

ces variables contiennent UN caractère. Ceux-ci sont classés suivant un ordre précis: le code ASCII qui suit l'ordre suivant :

- les chiffres '0' à '9' par ordre croissant
- les majuscules 'A' à 'Z' par ordre alphabétique
- les minuscules 'a' à 'z'

Dans le code ASCII, chaque caractère possible a un numéro de code. Par exemple A a pour code 65. En déclarant C comme variable caractère, on peut mettre le signe A dans C par C:='A' ou C:=CHR(65). Dans le premier cas, il faut mettre les cotes pour différencier 'A' de la variable A. Pour mettre une cote dans C, on peut faire C:=chr(39) ou C:="" : la 1ère cote pour dire qu'il va y avoir un caractère, les 2 suivantes qui symbolisent la cote (car une seule cote voudrait dire fin du caractère), la dernière qui signifie fin du caractère.

LES FONCTIONS STANDARD

On peut utiliser comme une variable des fonctions (qui peuvent soit être connues par le compilateur, soit définies par vous-même). Une fonction est un "module" ou "routine" qui renvoie une valeur au programme. Par exemple, $A := \text{sqrt}(B * C)$ met dans A la racine carrée de B fois C. B*C est appelé ARGUMENT de la fonction.

Les principales fonctions standard connues par tous les compilateurs sont :

ABS : renvoie la valeur absolue SIN : sinus
SQR : renvoie le carré ARCTAN : arc tangente
SQRT : racine carrée EX : exponentielle
COS : cosinus LN : log népérien
SUCC : variable énumérée suivante PRED : précédent
ROUND : arrondi à l'entier le plus proche
TRUNC : partie entière (permet de mettre un réel dans un entier:
trunc(4.5)=4)

Comme toute variable, une fonction possède un type (entier, réel,...) défini, et ne peut donc être utilisée que comme une variable de ce type.

INSTRUCTION

On appelle "instruction simple" soit :

- * une affectation
- * un appel à une procédure (une procédure est un ensemble d'instructions regroupées sous un nom, par exemple READLN)
- * une structure de contrôle (voir plus bas);

On appelle "instruction composée" le regroupement de plusieurs instructions sous la forme :

BEGIN instruction1;instruction2; ... ;instructionN END

On ne met pas de ; après BEGIN ni avant END (puisque le ; sépare deux instructions). Par contre, si l'instruction composée est SUIVIE d'une autre instruction, on mettra un ; après le END .

Rq La lisibilité du programme sera meilleure en mettant une instruction par ligne, et en décalant à droite les instructions comprises entre un begin et un end.

On appelle "instruction" une instruction soit simple soit composée

STRUCTURES DE CONTROLE

- [BOUCLE WHILE - DO \(tant que - faire\)](#)
- [BOUCLE REPEAT - UNTIL \(répéter - jusqu'à ce que\)](#)
- [BOUCLE FOR - DO \(pour - faire\)](#)
- [INSTRUCTION IF - THEN - ELSE \(si - alors - sinon\)](#)
- [LA STRUCTURE CASE - OF \(cas - parmi\)](#)

Nos connaissances actuelles ne nous permettent pas de faire des programmes utilisant la capacité de l'ordinateur de répéter rapidement et sans erreur beaucoup de calculs. Nous allons donc remédier immédiatement à cela. Chaque structure de contrôle forme une instruction (qui peut donc être utilisée dans une autre structure de contrôle).

BOUCLE WHILE - DO (tant que - faire)

structure : WHILE expression booléenne DO instruction

elle permet de répéter l'instruction tant que l'expression (ou la variable) booléenne est vraie.

```
ex :   PROGRAM racine_a_deux_decimales(input,output);
      VAR nombre,racine:REAL;
      begin
          writeln('entrez un réel entre 0 et 10');
          readln(nombre);
          racine:=0;
          WHILE racine*racine < nombre DO racine:=racine+0.01;
          writeln('la racine de ',nombre,' vaut à peu
près',racine)
      END.
```

Il faut noter que si l'expression booléenne est fausse dès le début, l'instruction n'est jamais exécutée (ici si NOMBRE=0). Attention, Pascal n'initialise pas automatiquement les variables à 0, c'est à dire que sans l'instruction RACINE:=0 le programme risquerait de donner une réponse fausse (RACINE valant n'importe quoi, il sera en général très supérieur à la racine cherchée).

On ne peut répéter qu'UNE seule instruction. Mais celle-ci peut être simple (comme dans l'exemple précédent) ou composée (begin - end).

EXERCICE (ex_puiss) faire un programme qui affiche les puissances de 2 jusqu'à une valeur maxi donnée par l'utilisateur (par multiplication successive par 2).

BOUCLE REPEAT - UNTIL (répéter - jusqu'à ce que)

```
structure :   REPEAT
              instruction1;
              instruction2;
              ...etc...
              instructionN
            UNTIL condition
```

Langage PASCAL

Les N instructions sont répétées jusqu'à ce que la condition soit vérifiée. Même si la condition est vraie dès le début, elles sont au moins exécutées une fois.

```
ex: PROGRAM jeu_simpliste(input,output);
    VAR a:integer;
    begin
        writeln('entrez le nombre 482');
        REPEAT
            readln(a)
        UNTIL a=482;
        writeln('c'est gentil de m'avoir obéi')
    end.
```

Quelle que soit la valeur initiale de A (même 482), la question sera au moins posée une fois (plus si vous désobéissez).

EXERCICE (ex_jeu) Faites un jeu qui demande de trouver le nombre entre 0 et 10 choisi par l'ordinateur (en comptant les coups). On utilisera la fonction RANDOM(N) (non standard, disponible en TURBO PASCAL) qui renvoie un entier entre 0 et N-1 compris, par l'instruction VALEUR_CHOISIE:=RANDOM(11).

BOUCLE FOR - DO (pour - faire)

structure :

FOR variable_énumérée:=valeur_début TO valeur_fin DO instruction

La variable_énumérée (non réelle) prend la valeur_début, et l'instruction est exécutée. Puis elle est incrémentée (on passe à la suivante, c.a.d si elle est entière on ajoute 1), et ce jusqu'à valeur_fin (compris).

L'instruction sera donc exécutée (valeur_fin - valeur_début + 1) fois. Si valeur_fin est inférieur à valeur_début l'instruction n'est jamais exécutée. Cette forme de boucle est utilisée chaque fois que l'on connaît le nombre de boucles à effectuer.

On peut utiliser un pas dégressif en remplaçant TO par DOWNTO.

```
ex: for lettre:='Z' downto 'A' do writeln(lettre)
```

écrit l'alphabet à l'envers(en déclarant LETTRE du type CHAR)

La variable_énumérée peut être utilisée (mais pas modifiée) dans l'instruction (simple ou composée). Elle est souvent appelée "indice" de la boucle. Sa valeur est perdue dès que l'on sort de la boucle.

EXERCICE (ex_moy) Faire un programme qui calcule la moyenne de N nombres. N doit être demandé par un READLN. (initialiser une variable à 0, y ajouter progressivement chaque note puis diviser par N).

INSTRUCTION IF - THEN - ELSE (si - alors - sinon)

structure : IF condition THEN instruction1 (CAS 1)

ou : IF condition THEN instruction1 ELSE instruction2 (CAS 2)

Si la condition est vraie, alors on exécute l'instruction1 (simple ou composée). Sinon, on passe à la suite (cas 1), ou on exécute l'instruction2 (cas 2).

Remarquez qu'il n'y a pas de ; devant le ELSE.

EXERCICE (ex_jeu_bis) modifier le jeu précédent (ex_jeu) en aidant le joueur (en précisant si c'est plus ou c'est moins).

L'instruction2 peut être composée ou entre autres être une instruction IF :

Langage PASCAL

```
IF condition1 THEN instruction1
ELSE IF condition2 THEN instruction2
ELSE IF condition3 THEN instruction3
.....
ELSE instructionN
```

Un ELSE correspond toujours au dernier IF rencontré (mais dont on n'a pas encore utilisé le ELSE).

```
ex: IF cond1 then if cond2 then inst1 {cond1 et cond2}
                        else inst2 {cond1 et pas cond2}
                        else if cond3 then inst3 {pas cond1 mais cond3}
                        else inst4 {ni cond1 ni cond3}
```

Si on désire autre chose, utiliser BEGIN et END :

```
IF cond1 THEN begin
                        if cond2 then inst1
                        end      (* le prochain ELSE se rapporte à COND1 puisque
l'instruction (composée) suivant THEN est terminée *)
ELSE inst2
```

LA STRUCTURE CASE - OF (*cas - parmi*)

Elle évite d'utiliser une trop grande suite de ELSE IF.

```
structure : CASE expression OF {regardez bien où j'ai mis les ;}
                liste_de_cas1:instruction1;
                liste_de_cas2:instruction2;
                .....
                liste_de_casN:instructionN
END
```

L'instruction i sera exécutée si l'expression appartient à la liste_de_cas i. Les autres ne seront pas exécutées (on passe directement au END). L'expression doit être de type énuméré (pas de réels). En TURBO PASCAL, on accepte une liste_de_cas particulière qui est ELSE (et doit être placée en dernier), pour prévoir le cas où expression n'appartient à aucun des cas cités au dessus. En MS-PASCAL on utilise de même OTHERWISE.

```
ex: CASE a*b OF  (* avec a et b déclarés entiers *)
    0 : writeln('un des nombres est nul');
    1,10,100,1000,10000 :
    writeln('le produit est une puissance de 10');
    (* 100000 est impossible en TURBO car supérieur à MAXINT
*)
END
```

Attention certains compilateurs n'acceptent pas de passer sur un CASE avec une valeur prévue dans aucune liste de cas.

EXERCICE (ex_calc) faire un programme simulant une calculatrice 4 opérations en utilisant CASE pour le choix de l'opération à effectuer.

TYPES ENUMERES NON STANDARDS

Si les types prédéfinis mentionnés précédemment ne vous suffisent pas, vous pouvez déclarer (donc créer) d'autres types. Il faut d'abord définir le type que vous désirez créer en donnant la liste ordonnée de toutes les valeurs possibles :

exemple :

TYPE tjour = (lundi, mardi, mercredi, jeudi, vendredi, samedi, dimanche);

Toutes les variables de ce type ne pourront pas avoir d'autres valeurs que celles que l'on a énumérées.

Il faut ensuite déclarer quelles variables seront de ce type :

ex: VAR jour1, jour2: tjour;

On peut alors les utiliser dans le programme grâce à :

* des affectations : JOUR1:=mercredi;

JOUR2:=JOUR1;

* des fonctions prédéfinies: PRED (précédent), SUCC (suivant), ORD (numéro d'ordre (dans la déclaration), entre 0 et N-1)

ex: pred(mardi)=lundi

succ(mardi)=mercredi

ord(mardi)=1

* des comparaisons : lundi<mercredi mais attention dimanche>lundi (suivant l'ordre donné dans la déclaration)

* des boucles : FOR jour1:=lundi TO vendredi DO ...

* des sélections de cas : CASE jour2 OF

lundi : ...;

samedi, dimanche : ...

END

Rq1: les types entiers, booléens, caractères sont des types énumérés prédéfinis. Toutes ces possibilités s'y appliquent donc (succ(10)=11, ord('A')=65,...). Par contre les réels ne sont pas énumérés (on ne peut pas définir de successeur), et ni CASE ni FOR ne peuvent s'y appliquer.

Rq2: READLN et WRITELN ne fonctionnent pas en standard pour les types définis par l'utilisateur, mais plusieurs compilateurs l'acceptent (pas TURBO).

LES TYPES INTERVALLES

On peut également définir un type énuméré comme sous-ensemble d'un autre type énuméré, en en donnant les bornes. En voici quelques exemples :

```
TYPE jourtravail=lundi..vendredi; {si on a déjà défini tjour }  
mois=1..12; {sous-ensemble du type INTEGER}  
byte=0..255; {prédéfini en TURBO, prend 1 octet en mémoire}  
minuscules='a'..'z';
```

Ces variables s'utilisent comme des variables du type dont est issu leur intervalle (on peut utiliser une variable MOIS

TABLEAUX

- [TABLEAUX UNIDIMENSIONNELS](#)
- [LES CHAINES DE CARACTERES](#)
- [TABLEAUX DE TABLEAUX](#)
- [TABLEAUX COMPACTES](#)

TABLEAUX UNIDIMENSIONNELS

On a souvent besoin de regrouper dans une seule variable (et donc un seul nom) plusieurs variables (exemple 3 coordonnées d'un même vecteur). On utilise pour cela les TABLEAUX. La manière la plus simple des les définir est :

```
VAR nom_tableau : ARRAY [type_index] OF type_composantes
```

```
ex: CONST dimension=3;
```

```
VAR vecteur1,vecteur2 : ARRAY [1..dimension] OF REAL
```

On peut utiliser le tableau complet (vecteur1:=vecteur2) pour l'affectation uniquement (vecteur1+vecteur2 est impossible directement). Mais on peut également accéder aux différentes composantes par un index (si vecteur1 est le vecteur unitaire porté par Y, alors VECTEUR1[1]=0 ,VECTEUR1[2]=1, VECTEUR1[3]=0).

l'index peut être variable:

```
PROGRAM heures(input,output);
TYPE tj=(lundi,mardi,mercredi,jeudi,vendredi,samedi,dimanche);
VAR jour:tj;
    nb_heures_cours:ARRAY[tj] OF integer;
begin
    {il faudrait ici entrer les nombres d'heures de cours par jour}
    FOR jour:=lundi TO samedi DO writeln(nb_heures_cours[jour])
end.
```

On peut également définir un TYPE tableau, par exemple :

```
TYPE typheures=ARRAY[tj] OF integer;
```

```
VAR nb_heures_cours:typheures;
```

EXERCICE (moy a) Ecrire le programme qui lit une liste de N nombres, calcule et affiche la moyenne puis l'écart entre chaque note et cette moyenne.

EXERCICE (rot.b) Ecrire le programme qui lit une liste de N nombres, la décale d'un cran vers le haut (le 1er doit se retrouver en dernier) l'affiche puis la décale vers le bas.

EXERCICE (clas.c) Classer automatiquement une liste de n entiers par ordre croissant puis décroissant.

LES CHAINES DE CARACTERES

On appelle chaîne de caractères une suite de caractères regroupés dans une seule variable. En PASCAL standard il faut la déclarer comme ARRAY[1..nbcars] OF CHAR. Les tests sont possibles entre chaînes (test suivant l'ordre ASCII du 1er caractère, puis automatiquement du 2ème, etc).

Langage PASCAL

ex : 'aaz'<'aba' 'aa' <'aaa' 'ZZZ'<'aaa' (majuscules)

On peut également utiliser ces tableaux dans READLN et WRITELN.

Mais en général, les compilateurs possèdent des types plus puissants, associés à des fonctions et procédures non standard. En Turbo on déclare une chaîne par STRING[nbcar]. Une fonction utile est alors LENGTH(variable_string) qui donne la longueur de la chaîne (inférieure ou égale à la dimension déclarée).

EXERCICE (str) Ecrire un programme qui détermine le nombre et les positions d'une chaîne dans une autre (ex ON dans FONCTION : en position 2 et 7).

TABLEAUX DE TABLEAUX

On peut faire des tableaux avec des composantes de tous types, y compris de tableaux. On peut voir une matrice comme un tableau de lignes qui sont elles même des tableaux de composantes (exemple matrice 5 lignes, 10 colonnes):

```
TYPE      ligne=array[1..10] of real;
          mat=array[1..5] of ligne;
```

l'écriture TYPE mat=array[1..5] of array[1..10] of real

ou même TYPE mat=array[1..5,1..10] of real

sont strictement équivalentes (sauf si la définition du type ligne est nécessaire pour d'autres variables).

soit la déclaration

```
VAR      m1,m2:mat;
          i,j:integer;
```

on peut écrire :

```
m1[i,j]:=10 {affectation de la valeur 10 en ligne i, colonne
j}
```

```
m1[i]:=m1[i+1] {copie complète de la ligne i+1 sur la ligne
i}
```

```
m2:=m1 {copie de la matrice complète }
```

mais il est impossible d'accéder directement à une colonne.

RAPPEL: READLN et WRITELN ne s'appliquent qu'à des variables simples prédéfinies (c.a.d les composantes, si celles-ci sont réelles, entières, chaînes ou booléennes).

EXERCICE (mat) faire le calcul de multiplication d'une matrice (M lignes, L col) par (L lignes, N col) (résultat M lignes, N col).

TABLEAUX COMPACTES

On peut en PASCAL standard définir des tableaux en PACKED ARRAY. Ils s'utilisent de la même manière que des tableaux normaux, mais prennent moins de place en mémoire. Sur certains ordinateurs, ceci se paie par un ralentissement du programme dans certaines applications. En TURBO ce n'est pas le cas, et elles sont donc automatiquement toutes compactées (inutile de le préciser).

Ex : VAR truc : PACKED ARRAY [1..10] OF typtruc;

On possède en standard les procédures de transfert :

PACK(table_non_packée , index_début , table_packée_resultat)

UNPACK(table_packée , table_non_packée_resultat , index_début)

pour transférer toute la table, mettre index_début à 1.

ENREGISTREMENTS

- [DECLARATION](#)
- [UTILISATION DES ENREGISTREMENTS](#)
- [LA STRUCTURE WITH - DO \(avec - faire\)](#)
- [ENREGISTREMENTS AVEC VARIANTES](#)

Dans un tableau, tous les constituants doivent être semblables. Ce n'est pas le cas des enregistrements, qui sont des variables composées de plusieurs variables (ou CHAMPS) de types différents.

DECLARATION

structure d'une déclaration d'enregistrement :

```
VAR nom_variable :RECORD champ1:type ; champ2:type ;.... END
```

ou

```
TYPE nom_type = RECORD champ1:type ; champ2:type ;.... END
```

ex: TYPE date = RECORD

```
    jour:1..31;
```

```
    mois:1..12;
```

```
    an:1980..1999
```

```
END;
```

```
facture = RECORD
```

```
    reference:integer;
```

```
    jour:date; {enregistrement d'enregistrement}
```

```
    client:string[100];
```

```
    total_HT:real
```

```
END;
```

```
VAR date1,date2:date;
```

```
    comptes:array[1..100] of facture;
```

```
    fact:facture;
```

Dans un enregistrement, chaque champ doit avoir un nom différent. Mais pour des enregistrements différents on peut réutiliser le même nom de champ (ici JOUR).

UTILISATION DES ENREGISTREMENTS

Comme l'affectation fonctionne toujours pour des variables de même type, on peut écrire DATE1:=DATE2.

On accède à une composante par NOM_VARIABLE . NOM_CHAMP :

```
writeln('référence de la facture ? ');
```

```
readln(fact.reference);
```

```
writeln('mois de la facture ? ');
```

```
readln(fact.jour.mois);
```

Langage PASCAL

```
writeln('jour de la facture ? ');  
readln(fact.jour.jour);
```

```
for i:= 1 to nb_fact do writeln(comptes[i].prix);
```

Une composante d'enregistrement s'utilise comme une variable du même type (avec les mêmes possibilités mais aussi les mêmes limitations).

LA STRUCTURE WITH - DO (avec - faire)

structure : WITH nom_enregistrement DO instruction

Elle permet d'omettre le nom_enregistrement dans l'instruction.

```
ex : FOR I:=1 TO nb_fact DO  
      WITH comptes[i] DO begin  
        reference:=i;      (* c.a.d comptes[i].reference *)  
        writeln('nom du client de la facture ndeg.',i,' ? ');  
        readln(client);  
      END
```

Dans un WITH on peut évidemment accéder à d'autres enregis-trements, il suffit de préciser le nom_enregistrement.champ. Si l'on utilise des WITH imbriqués, en cas d'ambiguïté des noms de champs on se réfère au WITH le plus proche (intérieur).

```
ex : WITH fact DO begin  
      WITH date1 DO begin  
        mois:=12;      {date1.mois}  
        reference:=1;  {fact.reference car date1.reference  
n'existe pas}  
        jour:=15;      {date1.jour car WITH DATE1 le plus à  
l'intérieur}  
        fact.jour:=date2  {fact.jour puisqu'on le précise}  
      end  
    end
```

ENREGISTREMENTS AVEC VARIANTES

Certains champs d'un enregistrement peuvent être différents suivant la valeur d'un des champs en utilisant la structure CASE OF dans la déclaration de l'enregistrement.

```
ex:  TYPE statut=(celibataire,marie,divorce,veuf);  
      perso=RECORD  
        nom:string[20];  
        CASE situation:statut OF  
          celibataire:();  
          marie:(enfants:0..10);  
          divorce,veuf:(enfants:0..10;remarie:boolean)  
        END;
```

Il ne peut y avoir qu'un CASE par enregistrement. Celui-ci doit se placer en dernier, après les champs commun (si il y en a). Mais dans ce CASE on peut avoir UN autre CASE imbriqué (etc). Les cas où il n'y a pas de champ variant doivent être précisés (ici celibataire). Le champ discriminant (situation) s'utilise comme un autre champ commun.
exemples d'instructions possibles : (x déclaré comme perso)

Langage PASCAL

```
x.situation:=marie;
```

```
if x.situation=veuf then writeln(x.enfants)
```

EXERCICE (tel) A l'aide d'un tableau de personnes (nom, prénom, numéro dans la rue, rue, département, ville, numéro de téléphone), faire un programme de recherche automatique de toutes les informations sur les personnes répondant à une valeur d'une rubrique donnée (tous les PATRICK , tous ceux de Saverne, etc...)

PROCEDURES ET FONCTIONS

- [GENERALITES](#)
- [PORTEE DES DECLARATIONS](#)
- [ARGUMENTS \(OU PARAMETRES\)](#)
- [LES FONCTIONS](#)
- [RECURSIVITE](#)

Nous avons déjà vu un certain nombre de procédures (ex WRITELN) et fonctions (SQRT, SIN...) prédéfinies par le compilateur. Mais si l'on en désire d'autres, il suffit de les définir.

GENERALITES

On peut regrouper un ensemble d'instructions sous un même nom. On forme alors un sous-programme ou procédure. On utilise les procédures :

- * chaque fois qu'une même suite d'instructions doit être répétée plusieurs fois dans un programme,
- * quand une suite d'instruction forme une action globale. Le programme est alors plus clair et les erreurs plus facilement détectables.

Pour pouvoir utiliser une procédure, il faut d'abord la déclarer. La déclaration des procédures et fonctions se fait après toutes les autres déclarations.

structure d'une entité de programme (routine) :

entête

déclaration des :

 labels; {pour les GOTO, déconseillé}

 constantes;

 types;

 variables;

définition des sous-routines; {sous-programmes}

BEGIN

 instructions

END

le programme principal comme les procédures et les fonctions ont toujours cette structure. L'entête est composée d'un mot clef (PROGRAM, PROCEDURE ou FUNCTION), suivi de l'identificateur (nom) de la routine, et de la liste des arguments entre parenthèses. Les arguments forment la liaison avec l'extérieur de la routine (clavier, écran et éventuellement fichiers pour le programme).

Langage PASCAL

PROGRAM remplir (output);	{entête du prog principal}
var i:integer;	{déclarations prog princ.}
	{dont déclaration LIGNE}
PROCEDURE ligne(n:integer);	{entête de la procédure}
var j:integer;	{déclarations procédure}
BEGIN	{corps de la procédure}
for j:=1 to n do write('*');	
writeln	
END;	
BEGIN	{instructions du prog princ}
for i:=1 to 25 do ligne(70)	
END.	

la procédure LIGNE écrit N caractères '*' sur une même ligne. Le programme remplit donc l'écran (25 lignes 70 colonnes) d'étoiles.

On peut appeler une procédure déclarée dans une routine n'importe où dans cette routine en indiquant simplement son nom comme si c'était une instruction. A l'appel d'une procédure, le programme interrompt son déroulement normal, exécute les instructions de la procédure, puis retourne au programme appelant et exécute l'instruction suivante. Tout ce passe donc comme si le nom de la procédure était remplacé dans le programme par les instructions de la procédure (avec n=70).

PORTEE DES DECLARATIONS

Celle-ci est symbolisée dans l'exemple ci-dessus par deux cadres : la variable I et la procédure LIGNE (avec un argument entier) sont déclarées dans REMPLIR, et donc connues dans tout le programme (rectangle extérieur). Par contre N et J sont déclarés dans LIGNE et ne sont connus (et utilisables) que dans le rectangle intérieur.

En d'autres termes :

* Une variable est LOCALE pour une procédure X si elle est déclarée dans X. Elle n'existe que dans X (et dans les procédures déclarées à l'intérieur de X). La routine qui comporte la procédure X ne peut donc pas accéder à cette variable locale.

Langage PASCAL

* Une variable est GLOBALE pour une procédure X si elle est déclarée dans une routine ENGLOBANT la procédure X. Elle peut être utilisée dans la procédure. La modifier la modifie également dans le routine appelante (englobante).

Si l'on avait déclaré une variable I dans la procédure LIGNE (au lieu de N ou J), celle-ci aurait été locale à la procédure, c'est à dire que, dans le programme principal, I désigne une autre case mémoire que dans la procédure. Modifier la variable locale I ne modifie pas la variable globale I (momentanément inaccessible).

Rq : Ceci s'applique à toutes les déclarations. En particulier, une procédure déclarée localement à l'intérieur d'une procédure est indéfinie à l'extérieur.

ARGUMENTS (OU PARAMETRES)

Les échanges d'informations entre une routine appelant une sous-routine peuvent se faire par l'intermédiaire des variables globales. Mais il est beaucoup plus intéressant d'utiliser les PARAMETRES :

```
Ex : PROGRAM machin (input,output);
      VAR a,b,c,d:real;
          PROCEDURE aff_somme(x,y:real);
              var z:real;
              begin
                  z:=x+y;
                  writeln(x , ' + ' , y , ' = ' , z)
              end;
      BEGIN
          { programme principal }
          writeln('entrez 4 valeurs : ');
          readln(a,b,c,d);
          aff_somme(a,b); aff_somme(3,5); aff_somme(c+a,d)
      END.
```

En appelant AFF_SOMME(A,B), la procédure prend pour X la valeur de A, et pour Y la valeur de B. On dit que les arguments sont "passés par valeur". Mais si la procédure modifiait X ou Y, A et B ne seraient pas modifiés dans le programme appelant. Pour répercuter les modifications des arguments, il faut les déclarer comme "variables" (ils sont alors dits "passés par adresse").

```
ex : procedure echange (VAR x,y:real);
      var z:real;
      begin
          z:=x;
          x:=y;
          y:=z
      end; {cette procédure échange les contenus des 2
arguments}
```

LES FONCTIONS

Tout ce qui a été dit pour les procédures s'applique également aux fonctions. La différence avec une procédure est qu'une fonction renvoie un résultat. L'entête est du type :

FUNCTION nom_fonction (liste_parametres):type_de_la_fonction

la liste des paramètres (en général passés par valeur) est de la même forme que pour une procédure, le type de la fonction étant le type du résultat retourné. On retourne le résultat par :

Langage PASCAL

NOM_FONCTION := ... Cette affectation ne peut se faire qu'une seule fois par appel à la fonction.

```
ex : program classer(input,output);
      var a,b,c:real;
      function MAX(x,y:real):real;
      begin
        if x>=y then MAX:=x else MAX:=y
      end;
      begin
        writeln('entrez deux valeurs : ');
        readln(a,b);
        c:=max(a,b);
        writeln('le plus grand est ',c)
      end.
```

La fonction MAX a 2 paramètres réels (X et Y) et renvoie un réel.

RECURSIVITE

C'est ainsi que l'on appelle le fait qu'une routine puisse s'appeler elle-même.

```
ex : function factorielle(n:integer):integer;
      begin
        if n<=1 then factorielle:=1
          else factorielle:=n*factorielle(n-1)
        end;
```

Par exemple en appelant factorielle(3), on calcule 3*factorielle(2). Or factorielle(2)=2*factorielle(1), qui lui vaut 1. Donc factorielle(3)=3*(2*1) (ce qui me paraît juste). Faites un petit dessin, à chaque appel on recrée de nouvelles variables locales, donc on obtient 3 cases N distinctes valant 3, 2 et 1, on les supprime petit à petit en passant sur le END.

Rq : Il faut toujours vérifier qu'en aucun cas on ne puisse avoir une boucle infinie qui bloquerait la machine. Ce serait le cas en mettant le test IF N=1 et en appelant factorielle pour une valeur négative ou nulle.

Une procédure devant toujours être déclarée pour pouvoir être utilisée, on utilise FORWARD pour les cas de récursivité passant par 2 procédures :

```
function prem(a,b:real):boolean; FORWARD;
{déclaration anticipée de l'entête }
procedure deux(x,y:real);
var bool:boolean;
begin
  .....
  bool:=prem(x,y); {on peut utiliser PREM car déjà déclarée}
  .....
end;
function prem;
{ne plus donner les arguments car déjà déclarés}
begin
  .....
  if pas_fini then deux(a,b);  (* DEUX déjà
déclarée*)
  .....
end;
```

Langage PASCAL

EXERCICE (rec) écrire le programme qui calcule le déterminant d'une matrice carrée $N \times N$ sachant que celui-ci vaut :

$$n$$

$$i=1$$

où $M[i,1]$ est l'élément de la matrice (ligne i , 1ère colonne),

DET_{n-1} est le déterminant de la sous-matrice d'ordre $n-1$ obtenu en ôtant la ligne i et la 1ère colonne.

Le déterminant d'une matrice 1×1 est son seul élément.

On utilisera bien évidemment une fonction récursive, et l'on séparera le calcul de sous-matrice dans une procédure.

Rq : Il existe des méthodes numériques permettant d'accéder au résultat beaucoup plus rapidement que par cette méthode.

LES ENTREES / SORTIES

- [SUR LA CONSOLE](#)
- [LES FICHIERS DE TEXTE](#)
- [EXTENSIONS NON STANDARD](#)
 - [ACCES DIRECT](#)
 - [IMPRIMANTE](#)
 - [AUTRES](#)

SUR LA CONSOLE

La procédure WRITELN permet d'afficher des résultats sur l'écran. Les différents arguments sont affichés les uns après les autres sur la même ligne. Le curseur est ensuite automatiquement mis en début de ligne suivante. Pour éviter ceci, on peut utiliser WRITE, qui s'utilise comme WRITELN mais le curseur est laissé derrière le dernier caractère. De même, READ fonctionne comme READLN, excepté le curseur qui reste sur la ligne.

WRITE et WRITELN acceptent le "formatage" : on peut imposer le nombre de caractères utilisés pour chaque variable. Soit I entier, R réel :

WRITE(I:5, ' ', R:7:2) écrira I sur 5 caractères, et R sur 7 caractères dont 2 après la virgule. Si les valeurs sont trop petites pour entrer dans le format, des blancs sont mis devant. Si elles sont trop grandes, le format est ignoré.

SUR FICHIER

Un fichier est un ensemble de données, écrites sur un "support" lisible par l'ordinateur (disquette, cartes perforées,...), et regroupées sous un nom. Un fichier peut contenir des caractères (fichier de textes), des programmes, des valeurs (fichier de données).

Etudions l'exemple suivant :

```
program recopier(input,output);
var fic_ent,fic_sor : file of real;
    x:real;
begin
    assign(fic_ent,'fichier1'); (* non standard *)
    reset(fic_ent);
    assign(fic_sor,'fichier2'); (* non standard *)
    rewrite(fic_sor);
    while not eof(fic_ent) do begin
        read(fic_ent,x);
        write(fic_sor,x)
    end;
    close(fic_ent);
    close(fic_sor)
end.
```

On déclare les fichiers par :

VAR nomfic : FILE OF type_du_contenu

Langage PASCAL

Un fichier peut contenir des éléments de n'importe quel type (aussi compliqué soit-il, même tableaux), mais tous les enregistrements du fichier doivent être du même type.

Avant d'utiliser le fichier, il faut relier son identificateur au nom effectif du fichier par ASSIGN. Ici, FIC_ENT correspondra à un fichier qui aura pour nom FICHER1 sur la disquette. ASSIGN n'est pas une fonction standard, d'autres compilateurs utilisent OPEN.

Puis il faut préciser si l'on va écrire ou lire sur le fichier par RESET ou REWRITE. Ceci positionne en début du fichier.

On écrit dans le fichier par WRITE(nomfic,liste_des_variables). Idem pour READ.

Il ne faut pas oublier de fermer le fichier (close) quand on n'en a plus besoin. CLOSE est nécessaire également si l'on veut refaire un nouvel ASSIGN sur le même fichier.

La fonction standard EOF(nomfic) est TRUE quand on arrive en fin de fichier (End Of File).

Rq : Lors de la définition initiale du Pascal, on devait ouvrir les fichiers avant d'appeler le programme. Les fichiers étaient des variables globales déclarées en arguments du programme (et donc passés du système d'exploitation au programme). Nous les déclarons maintenant DANS le programme, les deux seuls fichiers donnés en paramètres du programme restant INPUT et OUTPUT.

LES FICHIERS DE TEXTE

Le principal problème des fichiers précédents est que tous les enregistrements sont du même type. De plus ils ne sont pas directement imprimables ni visibles par un traitement de textes ou un programme écrit dans un autre langage. On y remédie par les fichiers texte (aussi appelés fichiers formatés), déclarés par :

VAR nomfic:TEXT

Ce sont des fichiers de caractères, et se comportent comme l'écran ou le clavier (qui sont d'ailleurs les fichiers text INPUT et OUTPUT utilisés automatiquement si on ne précise pas de nomfic dans les READ<ln> et WRITE<ln>). On a aussi les mêmes limitations (écriture d'entiers, réels et chaînes de caractères seulement).

```
ex : program lire(input,output,fic);
      type chaine=array[1..80]of char;
      var fic:text;
          ligne:chaine;
      begin
        assign(fic,'texte');
        rewrite(fic);
        writeln('tapez votre texte, il sera enregistré dans le
fichier TEXTE ');
        writeln('tapez FIN pour arrêter');
        repeat
          readln(ligne);
          writeln(fic,ligne)
        until ligne='FIN';
        close(fic)
      end.
```

EXTENSIONS NON STANDARD

ACCES DIRECT

La plupart des compilateurs acceptent la procédure `SEEK(nomfic,position)` qui permet de se positionner n'importe où dans le fichier (sans être obligé de lire dans l'ordre tous les éléments). Ceci n'est évidemment possible que si la taille des éléments est constante, donc pour tous les fichiers exceptés ceux du type `TEXT`, puisque les lignes n'y sont pas de longueur constante. L'accès à une information est donc beaucoup plus rapide (à condition de connaître sa position) On appelle ceci l'**ACCES DIRECT** à un fichier, par opposition à l' **ACCES SEQUENTIEL** prévu en standard. `SEEK` est d'autant plus utile que `RESET` autorise la lecture et l'écriture des fichiers à accès direct (`REWRITE` par contre efface tout le contenu du fichier, on ne peut donc plus qu'écrire).

IMPRIMANTE

Pour accéder à l'imprimante, deux solutions:

- * ouvrir un fichier `TEXT` sur disque, le remplir, et lorsque le programme est terminé le copier (par le `DOS`) sur imprimante
- * utiliser un nom de fichier prédéfini correspondant à l'imprimante. En Turbo, c'est `LST`. ex : `writeln(lst, 'COUCOU')`. `LST` n'a pas besoin d'être ouvert en `TURBO` (tout comme `OUTPUT`).

AUTRES

Turbo Pascal permet beaucoup d'autres opérations sur les fichiers (`Delete`, `Rename`, `Path...`) qui ne sont pas standard. Si vous les utilisez, rappelez vous qu'elles ne fonctionneront pas avec un autre compilateur (Microsoft par ex) ni sous un autre système d'exploitation (Unix par ex).

EXERCICE (fichier) Ecrire la procédure qui lit le fichier "annuaire" afin de rendre opérationnel l'exercice tel.

ENSEMBLES

Un ensemble est une "collection" d'éléments de même type (cf cours maths 6ème). Supposons vouloir représenter des vendeurs et leurs domaines d'action.

Le type ensemble est défini par SET OF :

```
TYPE produits=(velos,motos,autos,accessoires);  
VAR vendeur1,vendeur2 : SET OF produits;
```

On "remplit" un ensemble en donnant ses éléments entre crochets :

```
vendeur1:=[velos,motos];  
vendeur2:=[motos,accessoires];
```

l'ensemble vide est : []

On peut faire les opérations suivantes :

UNION : vendeur1+vendeur2=[velos,motos,accessoires]

INTERSECTION : vendeur1*vendeur2=[motos]

COMPLEMENT : vendeur1-vendeur2=[velos]

vendeur2-vendeur1=[accessoires]

les tests booléens possibles sont : = , <> , <= (inclus) , >= (contenant).

On teste l'appartenance d'un élément par IN : si X vaut motos, alors X IN VENDEUR1 et [motos]<=VENDEUR1 sont équivalents (IN compare un élément et un ensemble, alors que <= compare deux ensembles.

Rq: En général, on ne pense pas à utiliser les ensembles (le prof de maths aurait-il oublié de nous dire à quoi ça sert ?), et l'on s'embrouille dans des programmes complexes. Voici par exemple, des idées pour programmer facilement un automatisme défini par plusieurs Grafcet complexes ([xx] se lisant "ensemble des xx") :

```
[étapes actives]:=[étapes initiales]
```

```
pour chaque transition :
```

```
    si [étapes immédiatement précédentes] <= [étapes actives]  
    et [capteurs nécessaires] <= [capteurs]
```

```
    alors [ét. actives]:=[ét. actives]-[précédentes]+[suivantes]
```

```
activer [sorties] en fonction [étapes actives] et boucler
```

On peut trouver un exemple détaillé dans mon document sur la [mise en oeuvre du Grafcet](#)

POINTEURS

- [LES LISTES CHAINEES ET ARBRES](#)
- [LES POINTEURS EN PASCAL](#)

Un pointeur est une variable qui stocke l'adresse (c.a.d la position en mémoire) d'une variable.

LES LISTES CHAINEES ET ARBRES

Supposons une liste de 3 entiers. On suppose connaître pour chacun l'adresse du suivant :

```

+-----+      +-----+      +-----+
|  5  | ---+----> |  9  | ---+----> |  2  |
+-----+      +-----+      +-----+

```

si l'on veut insérer une valeur dans la liste, les modifications à apporter sont minimales :

```

+-----+      +-----+      +-----+
|  5  | ---+--+      |  9  | ---+----> |  2  |
+-----+      +-----+      +-----+
                v          ^
                +-----+ |
                |  2  | ---+--+
                +-----+

```

Contrairement à l'insertion dans un tableau, il est inutile de décaler les termes suivants. Pour connaître la liste, il suffit de connaître l'adresse du premier terme.

Pour représenter un arbre, il suffit pour chaque élément de connaître l'adresse de chaque fils :

```

                                +----+
                                | ---+-->
                                +-----+
                                | F1 | ---+-->
                                +-----+
                                | ---+-->
                                +----+
                                | ---+-->
                                +----+
                                | ---+-->
                                +-----+
                                | P-F i | ---+-->
                                +-----+
                                | F2 | ---+-->
                                +-----+
                                | ---+-->
                                +----+
                                | ---+-->
                                +----+
                                | ---+-->
                                +-----+
                                | F3 | ---+-->
                                +-----+
                                | ---+-->
                                +----+

```

Rq : si le nombre de fils n'est pas constant, on a intérêt à stocker uniquement le fils aîné, ainsi que le frère suivant.

LES POINTEURS EN PASCAL

En pascal, on utilise les pointeurs pour représenter ces objets. La déclaration se fait de la manière suivante :

```
TYPE pointeur=^type_de_la_variable_pointée
ex : TYPE tpoint=^tval;  (* pointe sur des TVAL *)
      tval=record
          valeur:integer;
          suivant:tpoint
      end;
      VAR p1,p2:tpoint;
```

Dans cet exemple, les variables de type TVAL contiendront un entier et l'adresse de la suivante (liste chaînée vue plus haut).

Contrairement aux tableaux, il n'est pas nécessaire de prévoir le nombre de variables pointées à l'avance. C'est "l'allocation dynamique de mémoire" : on réserve la place pour chaque variable en cours d'exécution du programme, par la commande NEW(nom_variable). On récupère la place, si la variable est devenue inutile, par DISPOSE(nom_variable).

P1 contient donc une adresse d'une variable de type TVAL. Cette variable sera P1^ (c.a.d pointée par P1). On la "remplit" donc par des affectations du type :

P1^.valeur:=15; P1^.suivant:=P2;

Examinons un programme qui lit puis affiche une liste chaînée d'entiers :

```
program liste(input,output);
TYPE tpoint=^tval;
      tval=record
          valeur:integer;
          suivant:tpoint
      end;
VAR prem,precedent,point:tpoint;
    i,n:integer;
begin
    write('combien d''éléments comporte votre liste ?');
    readln(n);
    new(prem);      (* le 1er est particulier : si on le perd, on
perd la liste *)
    write('1ère valeur ? ');
    readln(prem^.valeur);  (* lecture de l'enregistrement VALEUR
de la variable d'adresse (pointée par) PREM *)
    precedent:=prem;
    for i:=2 to n do begin
        new(point);      (* création d'une nouvelle variable *)
        write(i,'ième valeur ? ');
        readln(point^.valeur);
        precedent^.suivant:=point;  (* mise à jour du chaînage *)
        precedent:=point  (*se préparer pour la prochaine boucle*)
    end;
    precedent^.suivant:=NIL;
```

Langage PASCAL

```
(* NIL signifie "rien" car 0 est un entier, non une adresse *)
point:=prem;  (* heureusement, on se souvient du premier *)
for i:=1 to n do begin
  writeln(point^.valeur);
  point:=point^.suivant  (* pour la prochaine boucle *)
end
end.
```

EXERCICE (pointeurs) modifier ce programme pour qu'il permette de rajouter ou supprimer des éléments. Décomposez le en routines.

CORRECTION DES EXERCICES

-- Ex ex_tva

```
program tva(input,output);
var prix_unitaire, quantite,
    total_ht, tva, total_ttc:real;
begin
    writeln('prix de l''article ?');
    readln(prix_unitaire);
    writeln('quantité désirée ? ');
    readln(quantite);
    total_ht:=prix_unitaire*quantite;
    tva:=total_ht*(18.6/100);
    total_ttc:=total_ht+tva;
    writeln('total ht   : ',total_ht);
    writeln('tva       : ',tva);
    writeln('          -----');
    writeln('total ttc : ',total_ttc)
end.
```

-- Ex ex_puiss

```
program puissances(input,output);
var n,max:integer;
begin
    writeln('nombre maxi ? ');
    readln(max);
    n:=2;
    while n<=max do begin
        writeln(n);
        n:=n*2
    end;
    writeln('c''est fini')
end.
```

-- Ex ex_jeu

```
program jeu(input,output);
var choix,rep,nb:integer;
begin
```

Langage PASCAL

```
nb:=0;
choix:=random(11);
repeat
  nb:=nb+1;
  writeln('choix ndeg. ',nb,' ? ');
  readln(rep)
until rep=choix;
writeln('trouvé en ',nb,' coups')
end.
```

-- Ex ex_moy

```
program moyenne(input,output);
var n,i:integer;
    note,total,moyenne:real;
begin
  writeln('nombre notes à entrer ?');
  readln(n);
  total:=0;
  for i:=1 to n do begin
    writeln(i,'ième note ? ');
    readln(note);
    total:=total+note
  end;
  moyenne:=total/n;
  writeln('la moyenne est : ',moyenne)
end.
```

-- Ex ex_jeu_bis

```
program jeu_ameliore(input,output);
var choix,rep,nb:integer;
begin
  nb:=0;
  choix:=random(11);
  repeat
    nb:=nb+1;
    writeln('choix ndeg. ',nb,' ? ');
    readln(rep);
    if rep<choix then
      writeln('c''est plus')
    else if rep>choix then
      writeln('c''est moins')
    {le 2ème if empêche d'écrire si juste}
  until rep=choix;
  writeln('juste en ',nb,' coups')
end.
```

-- *Ex ex_calc*

```
program calculatrice(input,output);
var val1,val2,resultat:real;
    operation:char;
begin
    writeln('première valeur ?');
    readln(val1);
    writeln('opération (+ - * /) ? ');
    readln(operation)
    writeln('deuxième valeur ? ');
    readln(val2);
    case operation of
        '+':resultat:=val1+val2;
        '-':resultat:=val1-val2;
        '*':resultat:=val1*val2;
        '/':resultat:=val1/val2
    end;
    writeln('résultat : ',resultat)
end.
```

-- *EX moy.a*

```
program moyenne(input,output);
var n,compteur:integer
    somme,moyenne,ecart:real;
    note:array[1..100] of real;
begin
    repeat
        writeln('nb notes (100 maxi)?');
        readln(n)
    until (n>0) and (n<=100);
    {entrée notes et calcul de la somme}
    somme:=0;
    for compteur:=1 to n do
        begin
            writeln(compteur,'è note ?');
            readln(note[compteur]);
            somme:=somme+note[compteur]
        end;
    {calcul et affichage de la moyenne}
    moyenne:=somme/n;
    writeln('moyenne : ',moyenne);
    {calcul et affichage des écarts}
    writeln('écarts :');
    for compteur:=1 to n do
        begin
            ecart:=note[compteur]-moyenne;
            writeln(compteur,'ième note (',
                note[compteur],
```

Langage PASCAL

```
        ') : écart : ',ecart)
    end
end.
```

-- Ex rot.b

```
program rotation(input,output);
var index,n:integer;
    prem:real;
    tableau:array[1..100]of real;
begin
    repeat
        writeln('nb valeurs (100 maxi)?');
        readln(n)
    until (n>0) and (n<=100);
    (* entrée des valeurs *)
    for index:=1 to n do
        begin
            writeln(index,'ième valeur ?');
            readln(tableau[index]);
        end;
        writeln('on décale vers le haut');
        prem:=tableau[1]; {ne pas écraser!}
        for index:=2 to n do
            tableau[index-1]:=tableau[index];
        tableau[n]:=prem;
        for index:=1 to n do
            writeln(tableau[index]);
        writeln('on re-décale vers le bas');
        prem:=tableau[n];
        for index:=n downto 2 do
            tableau[index]:=tableau[index-1];
        tableau[1]:=prem;
        for index:=1 to n do
            writeln(tableau[index])
        end.
end.
```

-- Ex clas.c

```
program classer(input,output);
var n,i,index,petit,indexpetit:integer;
    avant,apres:array[1..100]of integer;
    pris:array[1..100] of boolean;
    {pour noter ceux déjà pris}
begin
    repeat
        writeln('nb valeurs (100 maxi) ?');
```

Langage PASCAL

```
    readln(n)
until (n>0) and (n<=100);
{entrée valeurs - initialisation de pris}
for index:=1 to n do begin
    writeln(index,'ième valeur ? ');
    readln(avant[index]);
    pris[index]:=false
end;
{ordre croissant,on cherche N valeurs}
for i:=1 to n do begin
    petit:=maxint; {plus grand possible}
{recherche du plus petit non pris}
    for index:=1 to n do
        if (not pris[index]) and
            (avant[index]<=petit) then begin
            petit:=avant[index];
            indexpetit:=index
        end;
    { sauvegarde dans le tableau APRES et
      mise à jour de PRIS }
    apres[i]:=petit;
    pris[indexpetit]:=true
end; { passage au prochain i }
{affichage du tableau APRES}
writeln('par ordre croissant : ');
for i:=1 to N do writeln(apres[i]);
{classement par ordre décroissant}
writeln('par ordre décroissant : ');
for i:=n downto 1 do writeln(apres[i])
    {n'auriez-vous pas tout refait ?}
end.
```

-- Ex str

```
program position(input,output);
var ch,sch:string[255];
    i,j,n,l,ls:integer;
begin
    writeln('chaîne à tester ? ');
    readln(ch);
    writeln('sous-chaîne à trouver ?');
    readln(sch);
    l:=length(ch);ls:=length(sch);
    n:=0;
    for i:=1 to l-ls do begin
        j:=1;
        while (j<=l)and(ch[i+j-1]=sch[j])
            do j:=j+1;
```


Langage PASCAL

```
    if j>ls then begin
        writeln('trouvé position ',i);
        n:=n+1
    end
end;
writeln(n,' fois ',sch,' dans ',ch)
end.
```

--Ex mat

```
program produit_mat(input,output);
var m1,m2,m3:array[1..10,1..10]of real;
    l,m,n,jl,jm,jn:integer;
begin
    writeln('nb lignes 1ère matrice ?');
    readln(m);
    writeln('nb colonnes 1ère matrice ?');
    readln(l);
    writeln('nb colonnes 2ème matrice ?');
    readln(n);
    (* entrée de m1 *)
    writeln('première matrice');
    for jm:=1 to m do for jl:=1 to l do
    begin
        writeln('lig',jm,', col',jl,'?');
        readln(m1[jm,jl])
    end;
    (* entrée de m2 *)
    writeln('2ème matrice');
    for jl:=1 to l do for jn:=1 to n do
    begin
        writeln('lig',jl,', col',jn,'?');
        readln(m2[jl,jn])
    end;
    (* calcul du produit *)
    for jm:=1 to m do for jn:=1 to n do
    begin {calcul composante m,n de m2}
        m3[jm,jn]:=0;
        for jl:=1 to l do m3[jm,jn]:=
            m3[jm,jn]+(m1[jm,jl]*m2[jl,jn]);
    end;
    (* affichage du résultat *)
    writeln('résultat');
    for jm:=1 to m do for jn:=1 to n do
        writeln('m['',jm,',',',jn,']= ',
            m3[jm,jn])
    end;
end.
```

-- *Ex tel*

```
program annuaire(input,output);
(* version simplifiée *)
type ligne=string[40];
   typepersonne=record
       nom:ligne;
       num_tel:ligne
   (* integer malheureusement < 32635 *)
   end;
var pers:array[1..100]of
       typepersonne;
    nb,i:1..100;
    rep:char;
    imprimer:boolean;
    texte:ligne;

begin
{on suppose avoir ici les instructions
permettant de lire sur fichier disque
NB et le tableau PERS }
    repeat
        writeln('recherche suivant : ');
        writeln(' N : nom');
        writeln(' T : numéro téléphone');
        writeln(' Q : quitter le prog');
        writeln('quel est votre choix ?');
        readln(rep);
        if rep<>'Q' then begin
            writeln('texte à chercher ? ');
            readln(texte)
            for i:=1 to nb do with pers[i] do
                begin
                    case rep of
                        'N':imprimer:=nom=texte;
                        'T':imprimer:=num_tel=texte;
                    end;
                    if imprimer then begin
                        writeln('nom   : ',nom);
                        writeln('tel   : ',num_tel)
                    end
                end
            end
            until rep='Q'
        end.
end.
```

-- *Ex rec*

```
program determ(input,output);
{ on se limite à 10x10, ce qui fait 7h
```

Langage PASCAL

```
de calcul et 6.235.314 appels à DETN }
type tmat=array[1..10,1..10] of real;
var dim:integer;
    {dimension matrice à calculer}
    det:real;    {résultat désiré}
    mat:tmat;    {matrice à calculer}
    appel:real; {nb appels à }

procedure entree;
var lig,col:integer;
begin
    writeln('dimension de la matrice ?');
    readln(dim); {DIM variable globale}
    writeln('entrez les composantes :');
    for lig:=1 to dim do begin
        writeln('pour la ligne ndeg. ',lig);
        for col:=1 to dim do begin
            writeln('colonne ',col,' ?');
            readln(mat[lig,col])
        end
    end
end;

procedure sous_mat(mdeb:tmat; var mfin:
                    tmat; ind,dim:integer);
{on supprime la colonne 1 et la ligne
 ind pour avoir la s/mat de dim-1}
var col,lig,l:integer;
begin
    l:=0;
    for lig:=1 to dim do begin
        if lig<>ind then begin
            l:=l+1;
            for col:=2 to dim do
                mfin[l,col-1]:=mdeb[lig,col]
            end
        end
    end
end;

function detn(m:tmat;d:integer):real;
{dét ordre d en fonction ordre d-1}
var result:real;
    mprim:tmat; {matrice intermédiaire}
    lig,signe:integer;
begin
    appel:=appel+1;
    if d=1 then detn:=m[1,1]
        (* fin de récursivité *)
    else begin
        result:=0;
        signe:=-1;
```

Langage PASCAL

```
    for lig:=1 to d do begin
        sous_mat(m,mprim,lig,d);
        signe:=-signe;
        {changer de signe à chaque ligne}
        result:=result +
            (signe*m[lig,1]*detn(mprim,d-1))
    end;
    detn:=result
end
end;

begin (* programme principal *)
    entree;
    appel:=0;
    det:=detn(mat,dim);
    writeln('résultat : ',det);
    writeln('nb appels DETN : ',appel)
end.
```

-- Ex fichier

```
procedure lirefic;
var i:1..100;
    f:file of typepersonne;
(* variables globales :
    NB et le tableau PERS *)
begin
    assign(f,'annuaire'); {non standard}
    reset(f);
    nb:=0;
    while not EOF(f) do begin
        nb:=nb+1;
        read(f,pers[nb])
    end;
    close(f)
end;
{à vous de faire la suite}
```

-- Ex pointeurs

```
program liste(input,output);
    TYPE tpoint=^tval;
        tval=record
            valeur:integer;
            suivant:tpoint
        end;
end;
```

Langage PASCAL

```
VAR prem:tpoint; {variable globale}
    n:integer;
    c:char;

procedure lire;
var precedent,point:tpoint;
    i:integer;
{ modifie N et PREM }
begin
    write('combien d''éléments?');
    readln(n);
    new(prem);
    write('1ère valeur ? ');
    readln(prem^.valeur);
    precedent:=prem;
    for i:=2 to n do begin
        new(point);
        write(i,'ième valeur ? ');
        readln(point^.valeur);
        precedent^.suivant:=point;
        precedent:=point
    end;
    precedent^.suivant:=NIL
(* le dernier ne pointe sur rien *)
end;

procedure afficher;
var point:tpoint;
    i:integer;
begin
    point:=prem;
    for i:=1 to n do begin
        writeln(point^.valeur);
        point:=point^.suivant
    end
end;

procedure supprimer;
var point,prec:tpoint;
    rep:char;
begin
    point:=prem;
    repeat
        write(point^.valeur,' à ôter ?');
        readln(rep);
        if rep='O' then begin
            n:=n-1;
            if point<>prem then begin
                prec^.suivant:=point^.suivant;
                dispose(point);
                point:=prec^.suivant
            end
        end
    until rep='N';
end;
```

Langage PASCAL

```
        (* se préparer pour la suite *)
        end
        else begin
            prem:=prem^.suivant;
            dispose(point);
            (* ancien premier *)
            point:=prem
        end
    end
    else begin
        (* pointer sur le suivant *)
        prec:=point;
        point:=point^.suivant
    end
until point=nil
end;

procedure rajouter;
var p1,p2,prec:tpoint;
    rep:char;
begin
    p1:=prem;
    repeat
        write(p1^.valeur,' rajouter un
élément avant (O/N) ? ');
        readln(rep);
        if rep='O' then begin
            n:=n+1;
            if p1<>prem then begin
                new(p2);
                write('valeur ? ');
                readln(p2^.valeur);
                prec^.suivant:=p2;
                p2^.suivant:=p1;
                prec:=p2;
            end
            else begin
                new(p1);
                write('valeur ? ');
                readln(p1^.valeur);
                p1^.suivant:=prem;
                prem:=p1
            end
        end
    end
    else begin
        (* pointer sur le suivant *)
        prec:=p1;
        p1:=p1^.suivant
    end
until p1=nil;
p1:=prec;
```

Langage PASCAL

```
    repeat
        write('ajouter un élément en fin
de liste (O/N) ? ');
        readln(rep);
        if rep='O' then begin
            n:=n+1;
            new(p2);
            write('valeur ? ');
            readln(p2^.valeur);
            p1^.suivant:=p2;
            p2^.suivant:=nil;
            p1:=p2
        end
    until rep<>'O'
end;

BEGIN {prog principal}
    lire;
    repeat
        writeln('A:afficher, S:supprimer R:rajouter, F:fin');
        write('votre choix ? ');
        readln(c);
        case c of
            'A':afficher;
            'S':supprimer;
            'R':rajouter
        end
    until c='F'
end.
```

Qui suis-je ?

Je m'appelle **Patrick TRAU**, j'ai près de 40 ans (mais du côté où l'on s'en éloigne), je suis [marié](#) et ai trois [enfants](#).

Je suis enseignant à l'ULP ([Université Louis Pasteur](#)) à Strasbourg, j'enseigne l'informatique, l'automatisme et la technologie, à l'IPST ([Institut Professionnel des Sciences et Technologies](#)). De formation technique (Bac E, [ENS Cachan](#) section [Mécanique](#), Agrégé de mécanique), j'ai effectué ma recherche au LMT ([Laboratoire de Mécanique et Technologie](#)) à Cachan, dans le groupe CAO. J'ai mis au point SYMATRAU, un SYstème expert de MAillage TRidimensionnel AUtomatique. J'y ai travaillé sur UNIVAC 1100, puis sur NorkData/Matra DataSystem, principalement en FORTRAN. Puis j'ai été nommé enseignant à l'Université de Metz, où j'ai enseigné l'informatique (Basic, Pascal) et l'automatisme, principalement en licence/maîtrise Technologie Mécanique, Génie Mécanique, préparation à l'agregation de mécanique, DESS GMP où je m'occupais du module CAO, et ISFATES. Au LPMM (Laboratoire de Physique et Mécanique des Matériaux), je m'occupais du matériel informatique, de leur installation physique à l'administration système du réseau. Je travaillais sur PC et VAX 780 puis sur stations Apollo sous AEGIS)

Depuis 1990 je suis en poste à l'IPST. Jusqu'en 94, j'étais directeur adjoint de l'IPST. Désormais je m'occupe de l'informatique de l'UFR, pour l'enseignement (PC), l'administration (PC / MAC) et la recherche au L3MI ([Laboratoire de Mécanique des Multi Matériaux Industriels](#)) où je gère la station SUN sous UNIX/Solaris 2. J'enseigne principalement l'informatique (C, PROLOG, Algorithmique) et [l'automatisme](#) en [IUP Génie des Systèmes Industriels](#) et [DESS Mécanique Avancée et Stratégies Industrielles](#),

Si vraiment vous désirez plus de détails sur mes activités, consultez mon [CV](#).

A part cela, il me reste un peu de temps pour m'occuper de ma famille, bricoler chez moi et sous ma voiture, faire du VTT avec des copains. Pendant les vacances, je pars en camping car, en général en France.